

Problem

Uruchomienie makra o określonym czasie

Słyszałem, że istnieje możliwość uruchomienia makra w Excelu o określonej porze. Mój problem polega na tym, że w firmie musimy uruchamiać ręcznie makro, które odświeża tabelę przedstawianą. Muszę to robić w godzinach 10:00–14:00 w odstępach co 15 minut (16 pomiarów, dzień w dzień) i wyniki z tabeli kopiować do komórek arkusza. Czy jest jakiś sposób na automatyzację tej pracy, tak aby Excel sam pamiętał, kiedy ma tę tabelę odświeżyć?

Rozwiązanie

W Excelu można skorzystać z metody *On.Time*, która doskonale sprawdza się w tego typu pracach. Pozwala ona na uruchomienie makra o konkretnej porze – ustalonej w sposób sztywny np. poprzez funkcję *TimeValue* (np. na godzinę 12:30), lub w sposób dynamiczny (np. 10 minut po uruchomieniu pliku).

Warto zaznaczyć, że plik nie musi być otwarty przed pierwszym uruchomieniem makra – Excel sam otworzy go o określonej porze.

Z drugiej strony, konieczne jest, aby przez cały czas otwarta była instancja programu – po zamknięciu Excela wszystkie zaplanowane uruchomienia zostaną wykasowane z pamięci.

Wracając do tematu. Rozwiązaniem może być makro, które przy otwarciu pliku (*Workbook_Open*) sprawdza bieżący czas i na jego podstawie określa pierwsze uruchomienie procedury *OdswiezDane*. Należy rozważyć tutaj trzy przypadki:

- Standardowo jeżeli plik został otwarty przed godziną 10:00 (ale po północy), makro powinno ustawić czas pierwszego uruchomienia na godzinę 10:00 (plus 5 minut zapasu na wypadek, gdyby komórka znajdowała się wówczas w trybie edycji).
- W sytuacji gdy plik został otwarty pomiędzy 10:00 a 10:05, chcemy, aby makro zostało uruchomione natychmiast.
- Jeżeli plik został otwarty po godz. 10:05 (ale przed północą), makro powinno się uruchomić automatycznie następnego dnia o godz. 10:00.

```
Private Sub Workbook_Open()
'Sprawdz aktualny czas i przypisz czas pierwszego uruchomienia
Select Case Time
  Case Is < TimeValue("10:00:00")
    gdteCzasUruchomienia = Date + TimeValue("10:00:00")
  Case TimeValue("10:00:00") To TimeValue("10:05:00")
    OdswiezDane: Exit Sub 'uruchom makro od razu
  Case Else
    gdteCzasUruchomienia = (Date + 1) + TimeValue("10:00:00")
End Select

'Zaplanuj pierwsze uruchomienia na 10:00 (nie dotyczy 10.00–10.05)
Application.OnTime EarliestTime:=gdteCzasUruchomienia, _
  Procedure:="OdswiezDane", _
  LatestTime:=gdteCzasUruchomienia + TimeValue("00:05:00")

'Wpisz czas następnego uruchomienia do komórki
Arkusz1.Range("K1").Value = gdteCzasUruchomienia
End Sub
```

Wiemy już, że makro *OdswiezDane* uruchomi się pierwszy raz o godzinie 10:00 (maksymalnie o 10:05). W makrze tym dodajemy do zmiennej publicznej *gdteCzasUruchomienia* 15 minut, następane uruchomienie nastąpi zatem o godz. 10:15

(maksymalnie o 10:20). Makro będzie uruchamiane aż do czasu, w którym zmienna *gdteCzasUruchomienia* osiągnie godzinę 14. Wtedy to zostanie uruchomiona procedura anulująca zaplanowane działanie makra.

```
Option Explicit
Public gdteCzasUruchomienia As Date

Sub OdswiezDane()
'Oblicz czas następnego uruchomienia - zapamiętaj wartość w komórce
gdteCzasUruchomienia = Now + TimeValue("00:15:00")
Arkusz1.Range("K1").Value = gdteCzasUruchomienia

'Zaplanuj następane uruchomienie makra
Application.OnTime EarliestTime:=gdteCzasUruchomienia, _
    Procedure:="OdswiezDane"

'Odswiez tabele przestawna i spisz wartości do tabeli
With Arkusz1
    .PivotTables("Tabela Przystawna1").PivotCache.Refresh
    .Range("A" & .Rows.Count).End(xlUp).Offset(1, 0).Resize(1, 8) = _
        WorksheetFunction.Transpose(.Range("J3:J10"))
End With
```

```
'Jeżeli czas następnego uruchomienia jest >=14:00 przerwij uruchamianie makr
If Hour(gdteCzasUruchomienia) = 14 Then
    Application.Run "AnulujMakro"
Exit Sub
End If
End Sub
```

```
Sub AnulujMakro()
'Anuluj działanie makra
Application.OnTime EarliestTime:=gdteCzasUruchomienia, _
    Procedure:="OdswiezDane", Schedule:=False

'Wyczyść komórkę z czasem następnego uruchomienia
Arkusz1.Range("K1").ClearContents

End Sub
```

Porady udzielił:

Mariusz Jankowski

autor strony internetowej poświęconej Excelowi i programowaniu w VBA;

Pytania: czytelnicy.controlling@infor.pl

Pobierz plik „Uruchomienie makra o określonym czasie” ze strony www.isc.infor.pl (zasoby płatne) – zakładka „Excel – pliki do samodzielnych ćwiczeń”.



UWAGA! Skoroszyt musi być otwarty od pierwszego do ostatniego uruchomienia makra (tj. w godz. 10:00–14:00). Ponowne otwarcie pliku w tych godzinach spowoduje, że Excel zaplanuje następane uruchomienie makra na godz. 10:00 następnego dnia. W tym celu należy zabezpieczyć się przed przypadkowym zamknięciem skoroszytu, np. poprzez obsługę zdarzenia *Workbook_BeforeClose*. Aby anulować programowe uruchomienie makra (zaplanowane w momencie otwarcia pliku), należy uruchomić ponownie MS Excel lub skorzystać z procedury *AnulujMakro*.